

双音多频（DTMF）信号发生器 HT9200A/B



作 者： 盛扬半导体（上海）有限公司软件部
日 期： 2001/8/6
适用芯片： HT9200A、HT9200B

简介

HT9200A/B 双音多频信号发生器被设计成 μC 界面。它们可由指令控制产生 16 种双频信号和 8 种单频信号，并由 DTMF 引脚输出。HT9200A 提供一种串行模式，而 HT9200B 可以选择为串行或并行模式，该芯片可以有多种应用，如安全系统、家庭自动化、通过电话线进行远程控制、通讯系统等。

HT9200A 的封装形式为 8DIP/SOP

HT9200B 的封装形式为 14DIP/SOP



串行模式（HT9200A/HT9200B）


在串行模式下，HT9200A/B 通过 DATA 引脚输入的一个 5bit 的代码来控制不同的 DTMF 信号输出，这 5 位代码按照 D0 到 D4 的顺序来传，并且数据要在 CLK 引脚下降沿到来之前放到输出锁存中。

控制代码与频率输出对应关系（串行模式）：

Digit	D4	D3	D2	D1	D0	频率输出（Hz）
1	0	0	0	0	1	697+1209
2	0	0	0	1	0	697+1336
3	0	0	0	1	1	697+1477
4	0	0	1	0	0	770+1209
5	0	0	1	0	1	770+1336
6	0	0	1	1	0	770+1477
7	0	0	1	1	1	852+1209
8	0	1	0	0	0	852+1336
9	0	1	0	0	1	852+1477
0	0	1	0	1	0	941+1336
*	0	1	0	1	1	941+1209
#	0	1	1	0	0	941+1477
A	0	1	1	0	1	697+1633
B	0	1	1	1	0	770+1633
C	0	1	1	1	1	852+1633

Digit	D4	D3	D2	D1	D0	频率输出 (Hz)
D	0	0	0	0	0	941+1633
—	1	0	0	0	0	697
—	1	0	0	0	1	770
—	1	0	0	1	0	852
—	1	0	0	1	1	941
—	1	0	1	0	0	1209
—	1	0	1	0	1	1336
—	1	0	1	1	0	1477
—	1	0	1	1	1	1633
DTMF OFF	1	1	1	1	1	—

时序图：



并行模式（只适用于 HT9200B）


HT9200B 除了能用于串行模式，还能用于并行模式。当 $\overline{S/P}$ 引脚为高电平时，HT9200B 工作于并行模式。在并行模式下，HT9200B 通过 D0~D3 引脚输入的的代码来控制不同的 DTMF 信号输出。HT9200B 是在 \overline{CE} 信号由高电平变为低电平时接收数据的。在 \overline{CE} 下降沿和 DTMF 产生信号输出之间会有一个延时（大约为 6ms）。

控制代码与频率输出对应关系（并行模式）：


Digit	D3	D2	D1	D0	频率输出 (Hz)
1	0	0	0	1	697+1209
2	0	0	1	0	697+1336
3	0	0	1	1	697+1477
4	0	1	0	0	770+1209
5	0	1	0	1	770+1336
6	0	1	1	0	770+1477
7	0	1	1	1	852+1209
8	1	0	0	0	852+1336
9	1	0	0	1	852+1477

Digit	D3	D2	D1	D0	频率输出 (Hz)
0	1	0	1	0	941+1336
*	1	0	1	1	941+1209
#	1	1	0	0	941+1477
A	1	1	0	1	697+1633
B	1	1	1	0	770+1633
C	1	1	1	1	852+1633
D	0	0	0	0	941+1633


时序图:



应用范例
应用电路
串行模式



并行模式



程序清单

```

; 文件名: HT9200.ASM
; 作 者: 姚晓峰
; 目 的: HT9200A/B 的使用介绍
include ht48c10.inc                ; 用 HT48C10 来控制 HT9200A/B
HT9200A      EQU 0                 ; 定义芯片型号
HT9200B      EQU 1                 ; 0 表示 HT9200A, 1 表示 HT9200B
SERIAL       EQU 0                 ; 工作模式定义
PARALLEL     EQU 1                 ; 0 表示串行模式, 1 表示并行模式

; -----
; 定义引脚
DATA         EQU PA.0
CE           EQU PA.1
CLK          EQU PA.2

; -----
SP           EQU PA.3
D0           EQU PB.0
D1           EQU PB.1
D2           EQU PB.2
D3           EQU PB.3

; -----
; 定义变量
data .section 'data'
count        db ?                  ; 可产生的 DTMF 信号数
buffer       db ?                  ; 数据传送缓存
bit_count    db ?                  ; 串行模式下控制数据传送位数
send_data    db ?                  ; 串行模式下数据传送缓存
delay_count1 db ?                  ; 延时变量 1
delay_count2 db ?                  ; 延时变量 2
delay_count3 db ?                  ; 延时变量 3

; -----
; 芯片型号和工作模式选择 (宏)
; TYPE 表示 HT9200A 或 HT9200B
; MODE 表示 SERIAL 或 PARALLEL
TYPE_MODE MACRO TYPE,MODE
IFE TYPE
    _TYPE EQU 0
    _MODE EQU 0
ELSE
    _TYPE EQU 1
IFE MODE
    _MODE EQU 0
ELSE
    _MODE EQU 1

```

```

ENDIF
ENDIF
ENDM

; -----
; 串行工作模式 (宏)
_SERIAL_MODE MACRO
    LOCAL  loop,next
    clr    pac
    clr    SP
    clr    CE
    set    CLK
    call   delay
loop:
    mov    a,18h
    mov    count,a
    clr    buffer
next:
    mov    a,buffer
    mov    send_data,a
    call   send_byte
    call   delay
    inc    buffer
    sdz    count
    jmp    next
    mov    a,0ffh
    mov    send_data,a
    call   send_byte
    call   delay
    jmp    loop
ENDM

; -----
; 并行工作模式 (宏)
_PARALLEL_MODE MACRO
    LOCAL  loop,next
    clr    pac
    clr    pbc
    set    SP
    clr    CE
loop:
    mov    a,10h
    mov    count,a
    clr    buffer
next:
    mov    a,buffer

```

; 循环产生 DTMF 信号
 ; 串行模式下有 24 种 DTMF 信号输出
 ; 产生下一个 DTMF 信号
 ; 控制代码传送
 ; 延时
 ; 关闭 DTMF 信号

; 设置 HT9200B 为并行模式
 ; 循环产生 DTMF 信号
 ; 并行模式下有 16 种 DTMF 信号输出
 ; 产生下一个 DTMF 信号

```

    mov    pb,a
    set    CE                      ; CE 下降沿传送控制代码
    clr    CE
    call   delay                  ; 延时
    inc    buffer
    sdz    count
    jmp    next
    jmp    loop
ENDM
; -----
code .section at 0 'code'
    org    00h
    jmp    start
    org    04h
    reti
    org    08h
    reti
; -----
send_byte proc                    ; 串行模式下的数据传送子程序
    mov    a,05h                  ; 传送 5 位数据
    mov    bit_count,a
next_bit:
    rrc    send_data
    sz     c
    jmp    send_1
send_0:                      ; 传送 0
    clr    DATA
    jmp    act
send_1:                      ; 传送 1
    set    DATA
act:                          ; CLK 下降沿传送数据
    clr    CLK
    set    CLK
    sdz    bit_count
    jmp    next_bit
    ret
send_byte endp
; -----
delay proc                      ; 延时子程序
    mov    a,02h
    mov    delay_count1,a
    mov    delay_count2,a
    mov    delay_count3,a
d: sdz    delay_count1

```

```

        jmp     d
        sdz     delay_count2
        jmp     d
        sdz     delay_count3
        jmp     d
        ret
delay endp
; -----
start:                                ; 主程序
;   TYPE_MODE HT9200A/HT9200B,SERIAL/PARALLEL
;   TYPE_MODE HT9200A,SERIAL          ; 芯片型号和工作模式选择
; 根据芯片型号和工作模式进行条件汇编
IF _TYPE
IF _MODE
    _PARALLEL_MODE
ELSE
    _SERIAL_MODE
ENDIF
ELSE
    _SERIAL_MODE
ENDIF
END

```

程序说明

在本应用中，我们采用 HT48C10 来控制 HT9200A/B 的 DTMF 发生。由于 DTMF 的产生有两种方式，串行模式和并行模式，所以我们定义了两个宏 `_SERIAL_MODE` 和 `_PARALLEL_MODE`，分别用于串行模式（SERIAL）和并行模式（PARALLEL）下产生 DTMF 信号。对于 HT9200A 来说，只有串行模式，而对于 HT9200B 而言则有两种模式，所以定义了一个宏 `TYPE_MODE`，可以根据芯片型号和工作模式进行条件汇编。